#-------------------------------------------------
#
# Project created by QtCreator 2013-09-21T21:52:09
#
#-------------------------------------------------

QT       += core gui opengl

greaterThan(QT_MAJOR_VERSION, 4): QT += widgets

TARGET = Plugin_SceneEditor
TEMPLATE = lib
CONFIG += plugin

CONFIG(debug, debug|release) {
    BUILD_DIR = $$PWD/../../Build/Tools/Debug
    TARGET = $$join(TARGET,,,_d)
    LIBS += -lQtExt_d
} else {
    BUILD_DIR = $$PWD/../../Build/Tools/Release
    LIBS += -lQtExt
}

LIBS += -L$$BUILD_DIR/lib

INCLUDEPATH += ./include
INCLUDEPATH += ../QtExt/include
INCLUDEPATH += ../Plugins/

contains(TEMPLATE, "lib") {
    DESTDIR = $$BUILD_DIR/lib
    DLLDESTDIR = $$BUILD_DIR/bin
} else {
    DESTDIR = $$BUILD_DIR/bin
}

UI_DIR = $$BUILD_DIR/build/ui/$$TARGET
OBJECTS_DIR = $$BUILD_DIR/build/obj/$$TARGET
MOC_DIR = $$BUILD_DIR/build/moc/$$TARGET
RCC_DIR = $$BUILD_DIR/build/rcc/$$TARGET


SOURCES += \
    source/SceneEditorPlugin.cpp \
    source/SceneEditorFrame.cpp \
    source/SceneEditor.cpp \
    source/SceneEditorDoc.cpp \
    source/SceneEditorView.cpp \
    source/QGridGraphicsItem.cpp \
    source/ScenePropertiesView.cpp \
    source/SceneImageLoader.cpp \
    source/ImageInfo.cpp \
    source/SceneImageSaver.cpp

HEADERS += \
    include/SceneEditorPlugin.h \
    include/SceneEditorFrame.h \
    ../Plugins/EditorWindow.h \
    ../Plugins/EditorPlugin.h \
    include/SceneEditor.h \
    include/SceneEditorDoc.h \
    include/SceneEditorView.h \
    include/QGridGraphicsItem.h \
    include/ScenePropertiesView.h \
    include/SceneImageLoader.h \
    include/ImageInfo.h \
    include/SceneImageSaver.h

OTHER_FILES += \
    json/SceneEditorPlugin.json

unix:!symbian {
    maemo5 {
        target.path = /opt/usr/lib
    } else {
        target.path = /usr/lib
    }
    INSTALLS += target
}

RESOURCES += \
    resource/SceneEditor.qrc

FORMS += \
    ui/SceneEditor.ui
